

# ANALITIČKA GEOMETRIJA RAVNINE I *MATHEMATICA*

MAJA BATOR, ZVONKO ČERIN I MILENA ČULAV

SAŽETAK. Opisuje se kako se *Mathematica* može upotrijebiti u analitičkoj geometriji ravnine s pravokutnim koordinatama. Kao ilustraciju mogućih primjena dana su rješenja petnaest zadataka iz zbirke za prvi razred gimnazije.

Ovaj članak opisuje kako se na računalu može obraditi analitička geometrija ravnine. Pritom se koristimo programom *Mathematica* koji je po svojim mogućnostima sličan programu *Maple V*. To su sigurno najpoznatiji i najrašireniji sustavi koji "znaju matematiku". Svaki od njih ima vlastiti programski jezik i naš zadatak svodi se na to da na tom jeziku opišemo nove funkcije koje trebamo za rješavanje geometrijskih zadataka analitičkom metodom.

Ranije smo u [1] i [2] to učinili u programu *Maple V*, a ovo je zapravo prilagodba tih članaka programu *Mathematica* tako da je osnovni tekst uglavnom ostao isti.

Na internet adresi <http://www.math.hr/~cerin> kao dodatak ovom članku pripremili smo *Mathematica* (verzija 4.2) bilježnicu (eng. notebook) sa svim inputima koji omogućuju kontrolu rezultata.


Ideja analitičke geometrije je da se geometrijska svojstva tvorevina opišu algebarskim izrazima te da se ona prouče metodama algebre. To se najjednostavnije postiže u ravnini pa ćemo od nje i krenuti.

Prvo moramo odlučiti kako ćemo algebarski prikazati **točku**, jer je ona osnovna tvorevina u geometriji ravnine. To se najlakše ostvaruje odabirom koordinatnog sustava. Poznato je više takvih sustava (npr. pravokutni, kosokutni, polarni, trilinearni, baricentrički,...), ali daleko najprirodniji je pravokutni koordinatni sustav pa ćemo ga i mi odabrati.

Taj sustav je zadan ako odaberemo dva orijentirana međusobno okomita pravca  $p_x$  i  $p_y$  i ako na jednom od njih (recimo na  $p_x$ ) odaberemo točku  $E$  različitu od točke njihovog presjeka  $O = p_x \cap p_y$ . Točku  $O$  nazivamo *ishodištem*, pravci  $p_x$  i  $p_y$  su  $x$ -os i  $y$ -os, a  $E$  je *jedinična točka* (vidi sliku 1). Položaj bilo koje točke  $T$  u ravnini jednoznačno je određen uređenim parom  $(x, y)$  realnih brojeva koji se zovu (pravokutne) *koordinate* točke  $T$ . Brojevi  $x$  i  $y$  su sljedeći omjeri orijentiranih dužina

$$x = \frac{|OT_x|}{|OE|}, \quad y = \frac{|OT_y|}{|OE|},$$

gdje su  $T_x$  i  $T_y$  ortogonalne projekcije točke  $T$  na  $x$ -os i na  $y$ -os.




SLIKA 1. Pravokutni koordinatni sustav.

Zadavanje točaka u ravnini u programu *Mathematica* je isto tako jednostavno budući da je točka zadana uređenim parom realnih brojeva. Na primjer, naredbom

`tA:={2, 3}; tB:={5, 7}; tC:={-2, 0}; tT:={x, y};`  
 zadali smo četiri točke ravnine  $A(2, 3)$ ,  $B(5, 7)$ ,  $C(-2, 0)$ ,  $T(x, y)$ .

Sada kada znamo zadavati točke ravnine opišimo kako se računa *udaljenost* među njima.



SLIKA 2. Udaljenost dviju točaka  $A(a, u)$  i  $B(b, v)$  u ravnini je  $|AB| = \sqrt{(b - a)^2 + (v - u)^2}$ .

Ako su  $A$  i  $B$  dvije točke u ravnini s koordinatama  $(a, u)$  i  $(b, v)$  vidimo da je udaljenost  $|AB|$  jednaka duljini hipotenuze pravokutnog trokuta  $ABC$ , gdje je  $C$  sjecište paralele s  $x$ -osi točkom  $A$  i paralele

s  $y$ -osi točkom  $B$ . Katete tog trokuta su (horizontalna)  $|b - a|$  i (vertikalna)  $|u - v|$  (vidi sliku 2). Zato je, prema Pitagorinom teoremu, ta udaljenost jednaka

$$|AB| = \sqrt{(b - a)^2 + (v - u)^2}.$$

U programu *Mathematica* udaljenost dviju točaka se računa pomoću sljedeće naredbe:


```
udaljenost2t[{a_,u_}, {b_,v_}]:=Sqrt[FS[(b-a)^2+(v-u)^2]]
```

Za ime te funkcije odabrali smo riječ *udaljenost2t* (skraćenica za "udaljenost dviju točaka"). Njene varijable su pak dva uređena para realnih brojeva kojima se prve i druge komponente zovu  $a$  i  $u$  odnosno  $b$  i  $v$ . Prvo stroj izračuna  $(b - a)^2 + (v - u)^2$  i onda pokušava što je više moguće pojednostavniti i rastaviti na faktore taj izraz (naredba *FS*). Na kraju nađe drugi korijen (naredba *Sqrt*) iz svega.

Da bi smanjili tipkanje tu smo uveli skraćenicu *FS* za istovremenu upotrebu naredbi *Factor* i *FullSimplify* koju ćemo često koristiti.

```
FS[m_]:=Factor[FullSimplify[m]]
```

U mnogim primjenama važno je odrediti *polovište dužine* čiji krajevi su poznati. Iz priložene slike 3 se vidi da su koordinate točke  $P$ , polovište dužine s krajevima u točkama  $A(a, u)$  i  $B(b, v)$ , jednake  $\frac{a+b}{2}$  i  $\frac{u+v}{2}$ .



SLIKA 3. Koordinate polovišta  $P$  dužine  $\overline{AB}$  su aritmetičke sredine koordinata krajeva.

Funkcija koja izračunava polovište dužine u programu *Mathematica* izgleda ovako:

```
poloviste[{a_,u_}, {b_,v_}]:=FS[{(a+b)/2,(u+v)/2}]
```

Nešto složenije je naći koordinate točke koja dijeli dužinu u zadanom omjeru. Dakle, tražimo točku  $C$  na pravcu  $AB$  takvu da je omjer duljina orijentiranih dužina  $\frac{|AC|}{|CB|}$  jednak zadanom broju  $\lambda$  (koji mora biti različit od  $-1$ ). Njene koordinate su  $\frac{a + \lambda b}{1 + \lambda}$  (apscisa) i  $\frac{u + \lambda v}{1 + \lambda}$  (ordinata). Najčešće se  $\lambda$  zadaje kao kvocijent  $\frac{m}{n}$  dvaju cijelih brojeva  $m$  i  $n$  (koji nisu jedan drugome suprotni).

Pripadne funkcije u programu *Mathematica* su:

```
omjerlambda[{a_,u_}, {b_,v_},k_] :=
  FS[{(a+k*b)/(1+k), (u+k*v)/(1+k)}]
```

```
omjermn[{a_,u_}, {b_,v_},m_,n_] :=
  FS[{(a*n+b*m)/(m+n), (u*n+v*m)/(m+n)}]
```

Polovište je, naravno, specijalan slučaj prethodnog postupka. Dobivamo ga iz općeg slučaja za  $\lambda = 1$  odnosno  $m = 1$  i  $n = 1$ .

Pored točaka u ravnini su najvažniji **pravci**. Prisjetimo se da se pravci zadaju linearnim jednadžbama. Točnije, ako su  $a$ ,  $b$  i  $c$  realni brojevi takvi da  $a$  i  $b$  nisu istovremeno nula, onda je skup svih točaka  $P(x, y)$  čije koordinate zadovoljavaju jednadžbu

$$ax + by + c = 0$$

pravac. Obrnuto, za svaki pravac mogu se naći takvi brojevi  $a$ ,  $b$  i  $c$  da koordinate svake točke tog pravca zadovoljavaju gornju jednadžbu.

Zato ćemo u programu *Mathematica* pravce zadavati uređenom trojkom  $\{a, b, c\}$  koeficijenata njegove linearne jednadžbe. Na primjer, naredbom

```
pY:={1, 0, 0}; pX:={0, 1, 0}; pD:={1, -1, 0}; pG:={-1, 2, 2}
```

zadali smo četiri pravca ravnine, i to  $y$ -os,  $x$ -os, simetralu prvog i trećeg kvadranta i pravac  $-x + 2y + 2 = 0$ .

Pravac se obično zadaje jednom svojom točkom  $P_0(x_0, y_0)$  i tangansom  $k$  kuta kojeg zatvara s pozitivnim smjerom  $x$ -osi. Njegova jednadžba je

$$y - y_0 = k(x - x_0),$$

pa ga u programu *Mathematica* zadajemo na sljedeći način:

```
pravackt[k_, {b1_,b2_}] := FS[{k, -1, b2-b1*k}]
```

Drugi najčešći način zadavanja pravca je pomoću njegove dvije različite točke  $P_1(x_1, y_1)$  i  $P_2(x_2, y_2)$ . Njegova jednadžba je

$$y - y_1 = \frac{y_2 - y_1}{x_2 - x_1} (x - x_1)$$

ako je  $x_1 \neq x_2$ . Napišemo li je u obliku

$$(y_2 - y_1)x + (x_1 - x_2)y + (x_1 y_2 - x_2 y_1) = 0$$

vidimo da vrijedi i ako je  $x_1 = x_2$ . Zato u ovom slučaju funkcija u programu *Mathematica* izgleda ovako:

```
pravac2t[{a1_,a2_},{b1_,b2_}] :=
  FS[{a2-b2,b1-a1,a1*b2-b1*a2}]
```

Ponekad je korisna i sljedeća funkcija kojom ispitujemo leži li dana točka na zadanom pravcu. Slovo Q u njenom imenu dolazi od engleske riječi "question" za pitanje. Točka leži na pravcu ako i samo ako je vrijednost izračunatog izraza jednaka nula.

```
tockanapravcuQ[{p_,q_},{a_,b_,c_}] := FS[a*p+b*q+c]
```

Sada lagano možemo naći uvjet koji moraju zadovoljavati koordinate triju točaka  $P_1(x_1, y_1)$ ,  $P_2(x_2, y_2)$  i  $P_3(x_3, y_3)$  da bi one ležale na istom pravcu (tj. da su kolinearne). Naime, gore smo vidjeli kako izgleda jednadžba pravca kroz točke  $P_1$  i  $P_2$ . Budući da točka  $P_3$  mora ležati upravo na tom pravcu njezine koordinate moraju zadovoljavati tu jednadžbu. Drugim riječima mora vrijediti

$$(y_2 - y_1)x_3 + (x_1 - x_2)y_3 + (x_1y_2 - x_2y_1) = 0.$$

To nam daje ideju kako definirati funkciju u programu *Mathematica* kojom ćemo određivati da li su tri zadane točke kolinearne ili ne.

```
kolinearneQ[{a1_,a2_},{b1_,b2_},{c1_,c2_}] :=
  FS[a1*b2-a1*c2-b1*a2+b1*c2+c1*a2-c1*b2]
```

Slično kao u slučaju funkcije `tockanapravcuQ` ovdje će tri točke biti kolinearne onda i samo onda kada je vrijednost funkcije `kolinearneQ` jednaka nuli.

Dva pravca  $A_1x + B_1y + C_1 = 0$  i  $A_2x + B_2y + C_2 = 0$  u ravnini se ili sijeku ili su paralelni. Ako se sijeku, onda su koordinate presjeka dane izrazima

$$x = \frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}, \quad y = \frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}.$$

Pretpostavka da se zadani pravci sijeku osigurava da su nazivnici u tim izrazima različiti od nule. Paralelni pravci nemaju presjek i

$$A_1 B_2 - A_2 B_1 = 0$$

je uvjet na njihove koeficijente koji mora biti ispunjen da bi oni bili paralelni. Prema tome, u programu *Mathematica* sljedećom funkcijom određujemo presjek dvaju pravaca u slučaju da oni nisu paralelni.

```
presjek2p[{a_,b_,c_},{u_,v_,w_}] :=
  FS[{(w*b-v*c)/(a*v-u*b), (u*c-a*w)/(a*v-u*b)}]
```

Da su paralelni znati ćemo ako nam primjenom te funkcije *Mathematica* javi grešku dijeljenja s nulom:

```
Power::"infy" : "Infinite expression  $\frac{1}{0}$  encountered."
```


Ako se pravci  $p_1$  i  $p_2$  s jednadžbama

$$A_1 x + B_1 y + C_1 = 0 \quad \text{i} \quad A_2 x + B_2 y + C_2 = 0$$

sijeku onda se tangens kuta  $\varphi$  među njima određuje formulom

$$\tan \varphi = \frac{A_1 B_2 - A_2 B_1}{A_1 A_2 + B_1 B_2} = \frac{k_2 - k_1}{1 + k_1 k_2},$$

gdje su  $k_1 = -\frac{A_1}{B_1}$  i  $k_2 = -\frac{A_2}{B_2}$  tangensi kuteva  $\varphi_1$  i  $\varphi_2$  što ih oni čine s  $x$ -osi (vidi sliku 4).



SLIKA 4. Kut između pravaca koji se sijeku.

Sljedeća funkcija to ostvaruje u programu *Mathematica*:

```
tankut2p[{a_, b_, c_}, {u_, v_, w_}] := FS[(v*a - u*b)/(u*a + v*b)]
```

Primjetimo da su pravci okomiti ako nam *Mathematica* javi grešku dijeljenja s nulom.

Sada vidimo da su pravci  $p_1$  i  $p_2$  paralelni ako je kut između njih jednak nuli ili ako je  $A_1 B_2 - A_2 B_1 = 0$ , odnosno  $k_1 = k_2$ . S druge strane oni su okomiti ako je  $A_1 A_2 + B_1 B_2 = 0$ , odnosno  $k_2 = -\frac{1}{k_1}$  (nazivnik gornjeg razlomka jednak je nula pa je kut pravi).

Te primjedbe opravdavaju sljedeće definicije funkcija koje traže paralelu danom pravcu kroz danu točku i okomicu na dani pravac kroz danu točku. Njima slične su one koje istražuju da li su dva pravca paralelna ili okomita.

```
paralela[{p_, q_}, {a_, b_, c_}] := FS[{a, b, -a*p - b*q}]
```

```
okomica[{p_, q_}, {a_, b_, c_}] := FS[{b, -a, a*q - b*p}]
```

```
paralelniQ[{a_, b_, c_}, {u_, v_, w_}] := FS[a*v - b*u]
```

`okomitiQ[{a_,b_,c_},{u_,v_,w_}] := FS[a*u+b*v]`

Ako nam funkcija `paralelniQ`, odnosno `okomitiQ`, za dane pravce, izračuna vrijednost nula, ti pravci su paralelni, odnosno okomiti.

Za tri različita pravca  $p_1$ ,  $p_2$  i  $p_3$  s jednadžbama

$$A_1 x + B_1 y + C_1 = 0, \quad A_2 x + B_2 y + C_2 = 0, \quad A_3 x + B_3 y + C_3 = 0$$

idemo odrediti uvjet što ga njihovi koeficijenti moraju zadovoljavati da bi oni bili kopunktalni ili konkurentni (tj. da su paralelni ili da se sijeku u jednoj točki).

Ako se pravci  $p_1$  i  $p_2$  sijeku u točki  $T(x, y)$  onda znamo da su  $x$  i  $y$  kvocijenti  $\frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}$  i  $\frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}$ . Uvrstimo li ih u  $p_3$  dobivamo

$$\frac{A_3 B_1 C_2 - A_3 B_2 C_1 + B_3 C_1 A_2 - B_3 C_2 A_1 + C_3 A_1 B_2 - C_3 A_2 B_1}{A_1 B_2 - A_2 B_1} = 0.$$

Dakle, ti pravci prolaze istom točkom jedino ako je izraz u brojniku

$$\Delta = A_3 (B_1 C_2 - B_2 C_1) + B_3 (C_1 A_2 - C_2 A_1) + C_3 (A_1 B_2 - A_2 B_1)$$

jednak nuli.

Ako su pravci  $p_1$ ,  $p_2$  i  $p_3$  paralelni onda su njihovi koeficijenti vezani relacijama  $A_2 = \lambda A_1$ ,  $B_2 = \lambda B_1$ ,  $A_3 = \mu A_1$ ,  $B_3 = \mu B_1$ , za neke realne brojeve  $\lambda$  i  $\mu$  različite od nule. Uvrstimo li te vrijednosti u  $\Delta$  opet se dobiva nula.

Obrnuto, ako je  $\Delta = 0$ , podijelimo li tu jednakost s  $A_1 B_2 - A_2 B_1$  vidimo da presjek  $T\left(\frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}, \frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}\right)$  pravaca  $p_1$  i  $p_2$  zadovoljava jednadžbu od  $p_3$ . Dakle, ta tri pravca su kopunktalna. To smo mogli načiniti jedino ako je  $A_1 B_2 - A_2 B_1 \neq 0$ , tj. ako  $p_1$  i  $p_2$  nisu paralelni.

Ako su  $p_1$  i  $p_2$  paralelni onda postoji realan broj  $\lambda$  takav da je  $A_2 = \lambda A_1$  i  $B_2 = \lambda B_1$ . Uvrstimo li to u izraz  $\Delta = 0$  dobijemo

$$(A_1 B_3 - A_3 B_1)(\lambda C_1 - C_2) = 0.$$


U produktu s lijeve strane druga zagrada nije nula jer je  $p_1 \neq p_2$ . Dakle,  $A_1 B_3 - A_3 B_1 = 0$  što znači da su  $p_1$  i  $p_3$  također paralelni i dokaz je potpun jer su onda  $p_1$ ,  $p_2$  i  $p_3$  paralelni.

Tako smo objasnili kako u programu *Mathematica* definirati sljedeću funkciju kojom ispitujemo da li su tri zadana pravca konkurentna.

`konkurentniQ[{a_,b_,c_},{i_,j_,k_},{p_,q_,r_}] :=  
FS[a*j*r-a*k*q-i*b*r+i*c*q+p*b*k-p*c*j]`

Dakle, tri pravca se sijeku u jednoj točki ako je vrijednost funkcije `konkurentniQ` jednaka nuli.

Pri rješavanju zadataka analitičke geometrije često treba odrediti ortogonalnu projekciju točke na pravac. Ta ortogonalna projekcija je presjek tog pravca i okomice na njega zadanom točkom. Dakle, ako napišemo



SLIKA 5. Refleksija točke u odnosu na točku (lijevo) i refleksija točke u odnosu na pravac (desno).

```
tQ:=presjek2p[m, okomica[tP, m]];
```

a zatim

```
tQ/.{tP->{p, q}, m->{a, b, c}}
```

program *Mathematica* će nam izračunati koordinate ortogonalne projekcije  $Q$  točke  $P(p, q)$  na pravac  $m$  s jednadžbom  $ax + by + c = 0$ . Zato pripadna funkcija izgleda ovako:

```
projekcija[{p_, q_}, {a_, b_, c_}] :=  
  FS[{(p*b^2 - a*c - a*b*q)/(b^2 + a^2),  
      (q*a^2 - b*c - a*b*p)/(b^2 + a^2)}]
```

Slično, mnogo puta trebamo naći refleksije točke u odnosu na neku točku ili neki pravac. Kada točku  $P$  reflektiramo u odnosu na točku  $Q$  koordinate reflektirane točke  $R$  mogu se odrediti ranije definiranom funkcijom *omjerlambda* za  $\lambda = -2$  zato jer točka  $R$  dijeli dužinu  $\overline{PQ}$  u omjeru  $-2$ . Ako napišemo

```
Clear["tP", "tQ"]; tR:=omjerlambda[tP, tQ, -2]
```

a zatim

```
tR/.{tP->{p, q}, tQ->{a, b}}
```

program *Mathematica* će nam izračunati koordinate refleksije točke  $P$  u odnosu na točku  $Q$ .

Zato je funkcija koja danim točkama  $P$  i  $Q$  pridružuje točku  $R$  (refleksiju točke  $P$  u odnosu na točku  $Q$ ) u programu *Mathematica* dana ovako:

```
refleksijaT[{p_, q_}, {a_, b_}] := FS[{2*a-p, 2*b-q}]
```

Slučaj refleksije točke  $P$  obzirom na pravac  $m$  je sličan. Sada je tražena točka  $R$  refleksija točke  $P$  u odnosu na ortogonalnu projekciju  $Q$  točke  $P$  na pravac  $m$ . Dakle, napišemo li sada

`tR/.{tP->{p,q},tQ-> projekcija[{p,q},{a,b,c}]}`

program *Mathematica* će izračunati koordinate tražene točke  $R$ . Prema tome pripadna funkcija ima sljedeći oblik:

`refleksijaP[{p_, q_},{a_, b_, c_}] :=  
FS[{(b^2*p-a^2*p-2*a*b*q-2*c*a)/(a^2+b^2),  
(a^2*q-b^2*q-2*a*b*p-2*c*b)/(a^2+b^2)}]`


Sada kada smo definirali osnovne funkcije analitičke geometrije ravnine, na petnaest primjera ćemo ilustrirati kako se one upotrebljavaju za rješavanje zadataka. Naš prvi primjer je zadatak 395. u knjizi [4] koji glasi ovako:

**Zadatak 1.** Dokaži da je površina  $P$  trokuta  $ABC$  s vrhovima u točkama  $A(x_1, y_1)$ ,  $B(x_2, y_2)$  i  $C(x_3, y_3)$  dana formulom:

$$P = \frac{1}{2} |x_1(y_2 - y_3) + x_2(y_3 - y_1) + x_3(y_1 - y_2)|$$

odnosno

$$P = \frac{1}{2} |y_1(x_2 - x_3) + y_2(x_3 - x_1) + y_3(x_1 - x_2)|.$$



SLIKA 6. Površina trokuta  $ABC$  je polovica umnoška duljina osnovice  $\overline{AB}$  i pripadne visine  $\overline{CD}$ .

*Rješenje.* Znamo da je površina trokuta jednaka polovici umnoška duljina bilo koje stranice i njoj pripadne visine. Pomoću funkcija programa *Mathematica* koje smo ranije definirali lagano izračunamo:

`tA:={Subscript[x,1],Subscript[y,1]};`

```

tB:={Subscript[x,2],Subscript[y,2]};
tC:={Subscript[x,3],Subscript[y,3]};
tD:=projekcija[tC,pravac2t[tA,tB]];
vP:=FS[udaljenost2t[tA,tB]*udaljenost2t[tC,tD]/2]

```

Nakom naredbe `vP` računalo će nam izračunati

$$\frac{1}{2} \sqrt{(x_1 - x_2)^2 + (y_1 - y_2)^2} \sqrt{\frac{(x_3(-y_1 + y_2) + x_2(y_1 - y_3) + x_1(-y_2 + y_3))^2}{(x_1 - x_2)^2 + (y_1 - y_2)^2}}.$$

Kako program *Mathematica* ipak ima nekih ograničenja u svojim mogućnostima vidimo da on ne skрати nazivnik u drugom kvadratnom korijenu s prvim kvadratnim korijenom (premda su oni očito jednaki), a isto tako ne primjećuje da je drugi korijen kvadrata u brojniku jednak apsolutnoj vrijednosti

$$|-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2|.$$

Kad načinimo ta skraćivanja očigledno dobijemo traženu formulu.  $\square$

Interesantno je da bez znaka apsolutne vrijednosti gornja formula daje izraz za orijentiranu površinu trokuta  $ABC$ . Ako je taj trokut pozitivno orijentiran, tj. ako je obilazak  $ABCA$  u smjeru suprotnom od smjera kazaljke na satu, onda će taj broj biti pozitivan, a inače je negativan. Funkcija za određivanje te, tzv. orijentirane površine trokuta, je u programu *Mathematica* ostvarena na sljedeći način:

```

povrsina3t[{a_, x_}, {b_, y_}, {c_, z_}] :=
  FS[(x*c-b*x-a*z+a*y+b*z-c*y)/2]

```

Naš drugi primjer je zadatak 425. iz iste knjige [4].

**Zadatak 2.** Neka je  $ABC$  trokut i  $P_1$ ,  $P_2$ ,  $P_3$  polovišta stranica  $\overline{BC}$ ,  $\overline{CA}$  i  $\overline{AB}$ . Dužine  $\overline{AP_1}$ ,  $\overline{BP_2}$  i  $\overline{CP_3}$  zovemo **težišnicama** trokuta  $ABC$ . Dokaži analitički da se sve tri težišnice sijeku u jednoj točki koju zovemo **težištem** trokuta i da težište dijeli svaku od težišnica u omjeru 2 : 1 računajući od vrha.

*Rješenje.* Dokaz na računalu, u programu *Mathematica*, počinje utipkavanjem sljedećeg:

```

tA:={Subscript[x,1],Subscript[y,1]};
tB:={Subscript[x,2],Subscript[y,2]};
tC:={Subscript[x,3],Subscript[y,3]};
tP1:=poloviste[tB,tC];tP2:=poloviste[tC,tA];
tP3:=poloviste[tA,tB]; konkurentniQ[pravac2t[tA,tP1],
  pravac2t[tB,tP2],pravac2t[tC,tP3]];


```

U nestvarno brzom vremenu računalo će ispisati vrijednost nula što pokazuje da se težišnice sijeku u jednoj točki. O kojoj se točki radi otkrivamo pomoću funkcije:

```

tT:=presjek2p[pravac2t[tA,tP1], pravac2t[tB,tP2]]

```



SLIKA 7. Trokut  $ABC$  i njegove težišnice  $\overline{AP_1}$ ,  $\overline{BP_2}$  i  $\overline{CP_3}$  koje se sijeku u težištu  $T$ .

Točka  $T$  ima koordinate  $(\frac{x_1+x_2+x_3}{3}, \frac{y_1+y_2+y_3}{3})$  pa odmah možemo napisati funkciju u programu *Mathematica* koja trokutu pridružuje njegovo težište.

```
teziste[{a_,x_},{b_,y_},{c_,z_}]:=
      FS[{(a+b+c)/3,(x+y+z)/3}]
```

Da bismo dokazali drugu tvrdnju problema mi ćemo odrediti točku koja dijeli težišnicu vrha  $A$  (tj. dužinu  $\overline{AP_1}$ ) u omjeru  $2 : 1$  računajući od vrha  $A$  i pokazati da se ona podudara s točkom  $T$  (težištem trokuta  $ABC$ ). Isti argument može se ponoviti i za težišnice vrhova  $B$  i  $C$ .

```
tG:=omjermn[tA,tP1,2,1]; udaljenost2t[tG,tT]
```

Budući da je dobivena vrijednost nula, točke  $G$  i  $T$  se podudaraju, pa je time zadatak riješen.  $\square$

Treći primjer je zadatak 989. također iz knjige [4].

**Zadatak 3.** Dokaži da polovišta stranica trokuta i nožišta visina leže na istoj kružnici.


*Rješenje.* Bez smanjenja općenitosti možemo pretpostaviti da su točke  $A$ ,  $B$  i  $C$  odabrane u ravnini tako da su im koordinate  $(0, 0)$ ,  $(c, 0)$  i  $(u, v)$ , gdje su  $c$ ,  $u$  i  $v$  realni brojevi, pri čemu  $c$  i  $v$  nisu jednaki nuli.

```
eA:={0, 0}; eB:={c, 0}; eC:={u, v}
```

Zatim odredimo polovišta stranica primjenom funkcije `poloviste`:

```
eAp:=poloviste[eB,eC]; eBp:=poloviste[eC,eA];
eCp:=poloviste[eA,eB]
```

Nožišta visina su ortogonalne projekcije vrhova na pravce nasuprotnih



SLIKA 8. Polovišta stranica i nožišta visina leže na istoj kružnici.

stranica:

```
eApp:=projekcija[eA,pravac2t[eB,eC]];
eBpp:=projekcija[eB,pravac2t[eC,eA]];
eCpp:=projekcija[eC,pravac2t[eA,eB]]
```

Središte trokutu opisane kružnice je točka presjeka simetrala njegovih stranica. Dakle, u našem slučaju je središte  $S$  opisane kružnice trokuta  $A'B'C'$  s vrhovima u polovištima stranica definirano ovako.

```
eS:=presjek2p[
    okomica[poloviste[eBp,eCp],pravac2t[eBp,eCp]],
    okomica[poloviste[eCp,eAp],pravac2t[eCp,eAp]]]
```

Primijenimo li isti postupak na trokut  $A''B''C''$  s vrhovima u nožištima visina možemo odrediti središte  $T$  njemu opisane kružnice.

```
eT:=presjek2p[
    okomica[poloviste[eBpp,eCpp],pravac2t[eBpp,eCpp]],
    okomica[poloviste[eCpp,eApp],pravac2t[eCpp,eApp]]]
```

Nakon utipkavanja naredbi  $eS$  i  $eT$  računalu nam ispisuje koordinate točaka  $S$  i  $T$ . Vidi se da su one jednake, pa su zato  $S$  i  $T$  jedna te ista točka.

Da bi dokazali tvrdnju zadatka moramo još pokazati da su radijusi opisanih kružnica trokuta  $A'B'C'$  i  $A''B''C''$  jednaki.

```
FS[udaljenost2t[eS,eCp]-udaljenost2t[eT,eCpp]]
```

Izračunata vrijednost je nula, čime je dokaz zadatka završen.

Interesantno je da možemo naći da je radijus gornje kružnice jednak polovici radijusa trokutu  $ABC$  opisane kružnice. Da bismo to provjerili prvo, koristeći gornje metode, definiramo koordinate točke  $O$  (središta

kružnice opisane trokutu  $ABC$ )

$e0 := \text{presjek2p}[\text{okomica}[eA, \text{pravac2t}[eB, eC]],$   
 $\text{okomica}[eB, \text{pravac2t}[eC, eA]]]$


a zatim zatražimo od programa *Mathematica* da izračuna sljedeće:

$\text{FS}[\text{udaljenost2t}[e0, eC] / \text{udaljenost2t}[eS, eCp]]$

Naravno, rezultat je dva. □

Četvrti primjer su zadatci 719. i 720. iz knjige [4].

**Zadatak 4.** Dokaži, ako su u trokutu dvije visine ili dvije težišnice jednake, onda je trokut jednakokračan.



SLIKA 9. Ako je  $|AA'| = |BB'|$  ili  $|AA''| = |BB''|$  onda je  $|BC| = |CA|$ .

*Rješenje.* Uz pretpostavke i oznake iz rješenja zadatka 3, utipkavanjem

$\text{FS}[\text{udaljenost2t}[eA, eApp]^2 - \text{udaljenost2t}[eB, eBpp]^2]$

dobivamo  $\frac{c^3 v^2 (2u - c)}{(v^2 + u^2 - 2uc + c^2)(v^2 + u^2)}$ . Dakle, ako su visine  $AA''$  i  $BB''$  jednake duljine mora biti  $u = \frac{c}{2}$  pa je  $ABC$  jednakokračan trokut.


Slično se vidi da unosom u program *Mathematica*

$\text{FS}[\text{udaljenost2t}[eA, eAp]^2 - \text{udaljenost2t}[eB, eBp]^2]$

dobijemo  $\frac{3c(2u - c)}{4}$  što vodi na isti zaključak za težišnice. □

Mnogo složeniji za dokazati je Zadatak 721. iz [4]. Naša metoda dokaza pretpostavlja poznavanje trigonometrijskih funkcija (posebno kotangensa).

**Zadatak 5.** Dokaži da je trokut jednakokračan onda i samo onda ako su mu dvije simetrale kuteva jednake.



SLIKA 10. Stranice  $BC$  i  $CA$  su jednake onda i samo onda ako su simetrale  $AA_i$  i  $BB_i$  kuteva  $A$  i  $B$  jednake.

*Rješenje.* Da bismo dobili što je moguće jednostavnije izraze pretpostaviti ćemo da točke  $A$  i  $B$  te središte  $I$  upisane kružnice trokuta  $ABC$  imaju koordinate  $(0, 0)$ ,  $(f + g, 0)$ , i  $(f, 1)$  gdje su  $f$  i  $g$  pozitivni realni brojevi. To su kotangensi polovica kuteva  $A$  i  $B$ .

$tA := \{0, 0\}$ ;  $tB := \{f+g, 0\}$ ;  $tI := \{f, 1\}$ ;  $tJc := \{f, 0\}$

Ako su  $J_a$ ,  $J_b$ ,  $J_c$  ortogonalne projekcije središta  $I$  upisane kružnice na stranice onda  $J_c$  ima koordinate  $(f, 0)$ , a koordinate od  $J_a$  dobivamo iz sistema jednadžbi

```
rjJa:=Solve[{udaljenost2t[tB,{p, q}]==
  udaljenost2t[tB,tJc], udaljenost2t[tI,{p, q}]==1},{p, q}];
```

gdje su  $p$  i  $q$  tražene koordinate točke  $J_a$ . Taj sistem ima dva rješenja i to koordinate točke  $J_c$  i koordinate  $\frac{f(g^2+1)+2g}{g^2+1}$  i  $\frac{2g^2}{g^2+1}$  od  $J_a$ .

```
tJa:={p,q} /. Extract[rjJa, 2]
```

Prisjetimo se da naredba `Extract[rjJa, 2]` izdvaja iz izraza za dva rješenja gornjeg sistema njegov drugi dio (što su zapravo koordinate točke  $J_a$ ).

Na sličan način se odrede i koordinate  $\frac{f(f^2-1)}{f^2+1}$  i  $\frac{2f^2}{f^2+1}$  točke  $J_b$ .

```
tJb:={p,q} /. Extract[Solve[{distance[tA,{p, q}]==
  distance[tA,tJc], distance[tI,{p, q}]==1},{p, q}], 2]
```

Sada možemo odrediti točke  $A_i$  i  $B_i$  u kojima simetrale kuteva  $A$  i  $B$  sijeku nasuprotne stranice kao presjeke  $AI \cap BJ_a$  i  $BI \cap AJ_b$ .

```
tAi:=presjek2p[pravac2t[tA,tI],pravac2t[tB,tJa]];
tBi:=presjek2p[pravac2t[tB,tI],pravac2t[tA,tJb]]
```

Zatražimo li od programa *Mathematica* da izračuna razliku kvadrata duljina simetrala kuteva utipkavanjem naredbe

$Q := \text{FS}[\text{udaljenost2t}[\text{tA}, \text{tAi}]^2 - \text{udaljenost2t}[\text{tB}, \text{tBi}]^2]$

kao izračun dobivamo

$$\frac{4(f+g)^3(f-g)(f^4g^2 + 4g^3f^3 - 5f^2g^2 + g^4f^2 + 4fg - 1)}{(g^2 + 2fg - 1)^2(f^2 + 2fg - 1)^2}.$$

Budući da brojnik sadrži  $f - g$  kao faktor i  $f + g$  očito nikada nije nula, vidimo da će dokaz biti gotov ako pokažemo da je dugačka zagrada

$$Z = f^4g^2 + 4g^3f^3 - 5f^2g^2 + g^4f^2 + 4fg - 1$$

u brojniku uvijek pozitivna.

I doista, ako su simetrale kuteva  $A$  i  $B$  jednake (tj.  $|AA_i| = |BB_i|$ ), onda je  $Q = 0$  i jer je  $(f+g)^3 \neq 0$  i  $Z \neq 0$  mora biti  $f - g = 0$  pa su kutevi  $A$  i  $B$  jednaki budući da su kotangensi njihovih polovica jednaki  $\cot \frac{A}{2} = f = g = \cot \frac{B}{2}$ . Kako u trokutu nasuprot jednakih kuteva leže jednake stranice slijedi da je  $|AC| = |BC|$  pa je trokut  $ABC$  jednakokrtačan. Obrnuto, ako je trokut  $ABC$  jednakokrtačan (tako da je  $|AC| = |BC|$ ), onda je  $f = g$  pa slijedi da je  $Q = 0$  i zato  $|AA_i| = |BB_i|$  (tj. simetrale kuteva  $A$  i  $B$  su jednake).

Prvo primjetimo da je zbroj  $\frac{A}{2} + \frac{B}{2}$  polovica kuteva najviše  $\frac{\pi}{2}$  pa je

$$\cot\left(\frac{A}{2} + \frac{B}{2}\right) = \frac{\cot(\frac{A}{2})\cot(\frac{B}{2}) - 1}{\cot(\frac{A}{2}) + \cot(\frac{B}{2})} = \frac{fg - 1}{f + g} > 0$$

što vodi na zaključak da je  $fg > 1$ .

Prvi i četvrti član od  $Z$  zajedno daju

$$f^4g^2 + f^2g^4 = (f^2 + g^2)(fg)^2 \geq 2(fg)(fg)^2 = 2(fg)^3$$

jer je  $f^2 + g^2 \geq 2fg$ . Uvedemo li oznaku  $\vartheta = fg$  vidimo da je

$$Z \geq 6\vartheta^3 - 5\vartheta^2 + 4\vartheta - 1.$$

Kako je  $\vartheta > 1$  zamjenimo li  $\vartheta$  u gornjem kubnom polinomu s  $1 + \tau$  dobiti ćemo  $(3\tau + 2)(2\tau^2 + 3\tau + 2)$  što je doista uvijek pozitivno jer je nova varijabla  $\tau = \vartheta - 1$  pozitivna pa je dokaz gotov.

Primjetimo da smo to isto mogli postići tako da u polinomu  $Z$  stavimo zamjenu  $f = \frac{1+k}{g}$  gdje je  $k$  pozitivan broj. Na naredbu

`Collect[Expand[Extract[Q,6] /. f->(1+k)/g], g,Factor]`


program *Mathematica* izračuna izraz

$$(1+k)^2g^2 + \frac{(1+k)^4}{g^2} + 2 + 6k + 7k^2 + 4k^3$$

koji je očigledno pozitivan. Prisjetimo se da naredba `Extract[Q,6]` izdvaja iz izraza  $Q$  njegov šesti dio (što je zapravo polinom  $Z$ ) dok nam funkcija `Collect` skuplja izraz `Extract[Q,6] /. f->(1+k)/g` (tj. polinom  $Z$  s uvrštenim  $\frac{1+k}{g}$  umjesto  $f$ ) po potencijama od  $g$  i još njegove koeficijente rastavlja na faktore.  $\square$

Nastavljamo sa zadatkom 833 iz [4] koji je iz odjeljka o sličnosti trokuta.

**Zadatak 6.** Neka je  $r$  polumjer trokutu  $ABC$  upisane, a  $R$  polumjer opisane kružnice. Dokaži da je  $R \geq 2r$ .



SLIKA 11. Polumjer opisane kružnice je veći ili jednak promjeru upisane kružnice.

*Rješenje.* Slijedeća metoda rješavanja ovog zadatka ima veliku sličnost s rješavanjem prethodnog zadatka. Bez smanjenja općenitosti možemo pretpostaviti da su kutevi  $A$  i  $B$  promatranog trokuta  $ABC$  oštri (tj. manji od  $\frac{\pi}{2}$  radijana) i da vrhovi  $A$ ,  $B$  i središte  $I$  upisane kružnice imaju koordinate  $(0, 0)$ ,  $(r(f+g), 0)$ , i  $(fr, r)$  za realne brojeve  $f > 1$ ,  $g > 1$ , i  $r > 0$ .

$tA := \{0, 0\}$ ;  $tB := \{r \cdot (f+g), 0\}$ ;  $tI := \{f \cdot r, r\}$ ;  $tJc := \{f \cdot r, 0\}$

Ideja dokaza je da nađemo koordinate vrha  $C$  i središta  $O$  opisane kružnice što će nam omogućiti da izračunamo radijus  $R$  opisane kružnice i da onda pokažemo da je razlika  $R - 2r$  uvijek veća od nule osim u slučaju jednakostraničnog trokuta kada je jednaka nuli.

Ako su  $J_a$ ,  $J_b$ ,  $J_c$  ortogonalne projekcije središta  $I$  upisane kružnice na stranice onda  $J_c$  ima koordinate  $(fr, 0)$ , a koordinate od  $J_a$  dobivamo iz sistema jednačbi

$rjJa := \text{Solve}[\{\text{udaljenost2t}[tB, \{p, q\}] == \text{udaljenost2t}[tB, tJc], \text{udaljenost2t}[tI, \{p, q\}] == r\}, \{p, q\}]$

gdje su  $p$  i  $q$  tražene koordinate točke  $J_a$ . Taj sistem ima dva rješenja i to koordinate točke  $J_c$  i koordinate  $\frac{(g^2 f + 2g + f)r}{1 + g^2}$  i  $\frac{2rg^2}{1 + g^2}$  od  $J_a$ .

$tJa := \{p, q\} /. \text{Extract}[rjJa, 2]$

Prisjetimo se ponovo da naredba `Extract[rjJa, 2]` izdvaja iz izraza za dva rješenja gornjeg sistema njegov drugi dio (što su zapravo koordinate točke  $J_a$ ).

Na sličan način se odrede i koordinate  $\frac{f(f^2-1)r}{f^2+1}$  i  $\frac{2f^2r}{f^2+1}$  točke  $J_b$ .

```
tJb:={p,q} /. Extract[Solve[{distance[tA,{p, q}]==
    distance[tA,tJc], distance[tI,{p, q}]==r},{p, q}], 2]
```

Sada možemo odrediti točku  $C$  kao presjek  $AJ_b \cap BJ_a$ .

```
tC:=presjek2p[pravac2t[tA,tJb],pravac2t[tB,tJa]]
```

Središte  $O$  opisane kružnice i njen radijus  $R$  dobivaju se iz slijedećeg sistema.

```
tO:={p, q}; Solve[{udaljenost2t[tA,tO]==R,
    udaljenost2t[tB,tO]==R, udaljenost2t[tC,tO]==R},{p, q, R}]
```

Od dva rješenja samo ono kod kojeg je  $R = \frac{r(1+g^2)(1+f^2)}{4(fg-1)}$  zadovoljava. U drugom rješenju je  $R$  negativan što ne možemo prihvatiti.

Razlika  $R - 2r$  jednaka je  $\frac{Mr}{4(fg-1)}$ , gdje je  $M$  kvadratni trinom

$$(g^2 + 1)f^2 - 8gf + g^2 + 9$$

u  $f$ .

```
R:=r*(1+f^2)*(1+g^2)/4/(f*g-1);
M:=Collect[Extract[FS[R-2*r],2],f];
\[CapitalDelta]:=FS[Coefficient[M,f,1]^2-
    4*Coefficient[M,f,2]*Coefficient[M,f,0]]
```

Njegova diskriminanta  $\Delta$  je  $-4(-3+g^2)^2$  što je uvijek negativno (pa je  $M > 0$  jer je  $g^2 + 1 > 0$ ) osim kada je  $g = \cot \frac{B}{2} = \sqrt{3}$  i zato  $f = \sqrt{3}$  (tj. trokut  $ABC$  je jednakostraničan) kada je  $M = 0$ .  $\square$

Slijedeći je zadatak 312. iz [4] koji je iz drugog poglavlja u dijelu o duljini kružnice i površini kruga.


**Zadatak 7.** Unutar trokuta  $ABC$  odabrana je točka  $T$  i neka su  $A_1, B_1, C_1$  redom bilo koje tri unutarnje točke na stranicama  $BC, CA, AB$  tog trokuta. Neka su dalje  $R_i, i = 1, 2, 3, 4, 5, 6$  redom polumjeri kružnica opisanih trokutima  $AC_1T, C_1BT, BA_1T, A_1CT, CB_1T, B_1AT$ . Dokaži da vrijedi  $R_1 R_3 R_5 = R_2 R_4 R_6$ .

*Rješenje.* Prvo potražimo funkciju koja će danoj trojci točaka pridruživati radijus opisane kružnice trokuta određenog tim točkama.

```
simetrala[a_, b_] := okomica[poloviste[a, b], pravac2t[a, b]];
sred0[a_, b_, c_] := presjek2p[simetrala[a, b], simetrala[a, c]];
rad0[a_, b_, c_] := udaljenost2t[a, sred0[a, b, c]]
```

Odredimo sada točke  $A, B, C$  i  $T$ .

```
Clear["s"]; tA:={0,0}; tB:={c,0}; tC:={s,t}; tT:={p,q}
```



SLIKA 12. Za radijuse šest opisanih kružnica vrijedi  $R_1 R_3 R_5 = R_2 R_4 R_6$ .

Ako je  $s \neq c$  onda se položaj bilo koje točke  $A_1$  na pravcu  $BC$  može odrediti nekim realnim brojem  $u$  i koordinate te točke su  $\left(u, \frac{t(c-u)}{c-s}\right)$ . To dobijemo tako da tražimo da točka s koordinatama  $(u, z)$  leži na pravcu  $BC$  i onda taj uvjet riješimo po varijabli  $z$ . Slično za  $s \neq 0$ , bilo koja točka  $B_1$  na pravcu  $CA$  ima koordinate  $(v, \frac{tv}{s})$ , a bilo koja točka  $C_1$  na pravcu  $AB$  ima koordinate  $(w, 0)$  za neke realne brojeve  $v$  i  $w$ .

```
Clear["s"]; tA1:={u, t*(c-u)/(c-s)}; tB1:={v, t*v/s};
tC1:={w, 0};
```

Ako je  $s = c$  onda bilo koja točka  $A_2$  pravca  $BC$  ima koordinate  $(c, u)$  za neki realni broj  $u$ . Ako je  $s = 0$  onda bilo koja točka  $B_2$  pravca  $CA$  ima koordinate  $(0, v)$  za neki realni broj  $v$ .

```
tA2:={c, u}; tB2:={0, v}
```

Odredimo sada funkciju koja računa razliku kvadrata produkata radijusa opisanih kružnica u odnosu na sedam točaka u ravnini.

```
FR[a_, b_, c_, d_, e_, f_, g_] := FS[(rad0[a, f, g]*rad0[b, d, g]*
rad0[c, e, g])^2 - (rad0[f, b, g]*rad0[d, c, g]*rad0[e, a, g])^2]
```

Sada se lagano provjeri da su slijedeće vrijednosti nula:

```
FR[tA, tB, tC, tA1, tB1, tC1, tT]
s:=c; FR[tA, tB, tC, tA2, tB1, tC1, tT]
s:=0, FR[tA, tB, tC, tA1, tB2, tC1, tT]
```


Tako smo Zadatak 312 iz [4] riješili u programu *Mathematica*.  $\square$

*Napomena 1.* Iz gornjeg dokaza je jasno da nigdje nismo koristili pretpostavku da je točka  $T$  unutar trokuta  $ABC$  niti pretpostavku da su

točke  $A_1$ ,  $B_1$ ,  $C_1$  unutarnje točke na stranicama  $BC$ ,  $CA$ ,  $AB$ . Na taj način smo upotrebom računala uspjeli dokazati mnogo općenitiju tvrdnju.

Slijedeći primjer je zadatak 644. iz zbirke [4] koji je iz odjeljka o volumenu valjka, stošca i kugle.

**Zadatak 8.** Na dnu valjkaste posude promjera 15 cm nalazi se kugla promjera 12 cm. U posudu je nalivena voda tako da njen nivo dotiče najvišu točku kugle. Za koliko će cm pasti nivo vode u posudi ako se kugla izvadi?



SLIKA 13. Valjkasta posuda sa kuglom u njoj.

*Rješenje.* Prisjetimo se formule  $V_K = \frac{4}{3} \left(\frac{D}{2}\right)^3 \pi$  za volumen kugle promjera  $D$  i formule  $V_V = \left(\frac{d}{2}\right)^2 h \pi$  za volumen valjka kojem je baza krug promjera  $d$ , a visina  $h$ .

U programu *Mathematica* funkcije volumena se definiraju ovako:

```
Vk[d_]:=d^3*Pi/6; Vv[d_, h_]:=d^2*h*Pi/4
```

Volumen vode u posudi je razlika volumena valjka i kugle

```
Vvoda:=Vv[15, 12]-Vk[12];
```

Nakon vađenja kugle voda se smjesti u valjak kojem je baza krug promjera 15 cm a visina mu je  $12 - p$  cm gdje je  $p$  traženi pad nivoa


vode u posudi. Zato se  $p$  nađe u programu *Mathematica* ovako:

```
N[p /. Part[Solve[Vvoda == Vv[15, 12 - p], p], 1]]
```

Dobije se vrijednost  $p = 5.12$  cm.  $\square$

Još jedan lijepi primjer je Zadatak 963. iz [4]. I ovdje se pretpostavlja poznavanje trigonometrijskih funkcija.

**Zadatak 9.** Oko kružnice radijusa  $R$  opisan je trapez. Tetiva koja spaja dirališta na krakovima paralelna je osnovicama i ima duljinu  $b$ . Dokaži da je površina trapeza  $P = \frac{8R^3}{b}$ .



SLIKA 14. Površina trapeza  $ACEG$  je  $\frac{8R^3}{b}$ .

*Rješenje.* Odaberimo pravokutni koordinatni sustav tako da kružnica  $k$  radijusa  $R$  koja je upisana promatranom trapezu  $ACEG$  ima središte u ishodištu, a njegove paralelne stranice  $AC$  i  $EG$  dodiruju  $k$  u točkama  $B(0, -R)$  i  $F(0, R)$ . Neka krakovi  $CE$  i  $AG$  dodiruju  $k$  u točkama  $D(R \cos \theta, R \sin \theta)$  i  $H(R \cos \sigma, R \sin \sigma)$  za neke kuteve  $\theta$  i  $\sigma$ .

Zadajmo u programu *Mathematica* prvo točke  $O$ ,  $B$ ,  $F$ ,  $D$ ,  $H$  i pravce  $AC$ ,  $EG$ .

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R};
tD:={R Cos[Theta], R Sin[Theta]};
tH:={R Cos[Sigma], R Sin[Sigma]};
pAC:={0, 1, R}; pEG:={0, 1, -R}
```

Prvo tražimo kada će spojnica  $DH$  dirališta krakova biti paralelna s osnovicama.

```
paralelniQ[pravac2t[tD, tH], pAC]
```

Uvjet je  $R(\sin \theta - \sin \sigma) = 0$  pa mora biti  $\sigma = \pi - \theta$ . Dakle, trapez  $ACEG$  je istokračan i simetričan u odnosu na pravac  $BF$  pa je dovoljno odrediti površinu desne polovice  $BCEF$ .

Pravac  $CE$  je okomica u točki  $D$  na pravac  $OD$ , a točke  $C$  i  $E$  su presjeci pravca  $CE$  s pravcima  $AC$  i  $EG$ .

```
pCE:=okomica[tD, pravac2t[t0, tD]];
tC:=presjek2p[pAC, pCE]; tE:=presjek2p[pEG, pCE]
```

Površina desne polovice  $BCEF$  je zbroj površina trokuta  $BCE$  i  $BEF$ .

```
povrsina3t[tB, tC, tE]+povrsina3t[tB, tE, tF]
```

Program *Mathematica* će nam za tu sumu izračunati vrijednost  $\frac{2R^2}{\cos \theta}$ . Kako je  $b = 2R \cos \theta$  vidimo da je tražena površina trapeza  $ACEG$  doista  $\frac{8R^3}{b}$ .  $\square$


*Napomena 2.* U knjizi [4] krivo se tvrdi da je površina trapeza  $\frac{4R^3}{b}$ .

Koristeći pristup iz rješenja Zadatka 13 (tj. zadatka 1112 u [4]) može se u potpunosti izbjeći upotreba trigonometrijskih funkcija. To rješenje prepuštamo čitateljima.

Nastavljamo s rješavanjem zadatka 1026. opet iz zbirke [4].

**Zadatak 10.** Dokaži da u pravilnom sedmerokutu  $A_1A_2A_3A_4A_5A_6A_7$  vrijedi:

$$\frac{1}{|A_1A_2|} = \frac{1}{|A_1A_3|} + \frac{1}{|A_1A_4|}.$$



SLIKA 15. U pravilnom sedmerokutu vrijedi:  $\frac{1}{|A_1A_2|} = \frac{1}{|A_1A_3|} + \frac{1}{|A_1A_4|}$ .

*Rješenje.* Odaberimo pravokutni koordinatni sustav tako da kružnica  $k$  sa središtem u ishodištu i radijusom  $R$  bude opisana promatranom pravilnom sedmerokutu. Možemo pretpostaviti da vrh  $A_1$  ima koordinate  $(R, 0)$ . Preostali, za nas značajni, vrhovi imaju koordinate  $A_2 (R \cos \frac{2\pi}{7}, R \sin \frac{2\pi}{7})$ ,  $A_3 (R \cos \frac{4\pi}{7}, R \sin \frac{4\pi}{7})$ ,  $A_4 (R \cos \frac{6\pi}{7}, R \sin \frac{6\pi}{7})$ .

Zadajmo u programu *Mathematica* te točke:

```
tA1:={R, 0};
tA2:={R Cos[2 Pi/7], R Sin[2 Pi/7]};
tA3:={R Cos[4 Pi/7], R Sin[4 Pi/7]};
tA4:={R Cos[6 Pi/7], R Sin[6 Pi/7]}
```

Da bismo provjerili relaciju među recipročnim vrijednostima utipkamo u program *Mathematica* slijedeće:


```
FullSimplify[Numerator[Together[1/udaljenost2t[tA1, tA2]
-1/udaljenost2t[tA1, tA3]-1/udaljenost2t[tA1, tA4]]], R>0]
```

Za nekoliko sekundi računalo će nas izvijestiti da je vrijednost jednaka nuli što dokazuje da tvrdnja zadatka vrijedi.  $\square$

*Napomena 3.* Neka druga interesantna svojstva pravilnih sedmerokuta dokazana u programu *Maple V* mogu se naći u autorovom članku [3].

Slijedeći je zadatak 1084. iz osmog paragrafa zbirke [4].

**Zadatak 11.** Projekcije kateta pravokutnog trokuta na hipotenuzu imaju duljine  $\frac{18}{5}$ ,  $\frac{32}{5}$ . Koliki je polumjer kružnice upisane u taj trokut?



SLIKA 16. Projekcije  $AD$  i  $DB$  kateta su zadane, a traži se radijus  $r$  upisane kružnice.

*Rješenje.* Smjestimo pravokutni koordinatni sustav tako da je njegovo ishodište vrh promatranog pravokutnog trokuta, a katete mu leže na koordinatnim osima. Možemo pretpostaviti da preostali vrhovi  $A$  i  $B$  ima koordinate  $(0, b)$  i  $(a, 0)$ , za neke pozitivne realne brojeve  $a$  i  $b$ .

U programu *Mathematica* te točke zadajemo ovako:

```
tO:={0, 0}; tA:={0, b}; tB:={a, 0}
```

Sada nademo ortogonalnu projekciju  $D$  vrha  $O$  na hipotenuzu  $AB$ .

$tD := \text{projekcija}[t0, \text{pravac2t}[tA, tB]]$

Vrijednosti parametara  $a$  i  $b$  možemo odrediti iz informacije da je  $|AD| = \frac{18}{5}$  i  $|BD| = \frac{32}{5}$ .

$\text{Solve}[\{\text{udaljenost2t}[tA, tD] == 18/5, \\ \text{udaljenost2t}[tB, tD] == 32/5\}, \{a, b\}]$

Ima čak osam rješenja (četiri realna i četiri imaginarna) ali samo jedno  $a = 8$  i  $b = 6$  nam odgovara. Dakle, promatrani pravokutni trokut ima stranice 8, 6, 10 (koje su dvostruko dulje od standardnog pravokutnog trokuta 4, 3, 5) pa mu upisana kružnica ima radijus  $r = 2$ .


To možemo vidjeti i tako da tražimo da središte  $I$  upisane kružnice s koordinatama  $(r, r)$  bude za  $r$  udaljeno od pravca  $AB$ .

$a := 8; b := 6; tI := \{r, r\}; \text{Solve}[\text{udaljenost2t}[tI, \\ \text{projekcija}[tI, \text{pravac2t}[tA, tB]]] == r, r]$

Od dva rješenja  $r = 2$  i  $r = 12$  jedino prvo zadovoljava uvjete zadatka.  $\square$

Sada razmatramo zadatak 1103. opet iz osmog paragrafa zbirke [4].

**Zadatak 12.** Dvije stranice trokuta imaju duljine 6 cm i 8 cm. Težišnice tih stranica sijeku se pod pravim kutem. Odredi treću stranicu trokuta.



SLIKA 17. Poznate su stranice  $a$  i  $b$  i težišnice  $AT_a$  i  $BT_b$  su okomite, a traži se stranica  $c$ .

*Rješenje.* Neka trokut  $ABC$  bude smješten u pravokutni koordinatni sustav tako da je  $A(0, 0)$ ,  $B(c, 0)$ , i  $C(u, v)$  za neke pozitivne realne brojeve  $c$  i  $v$  i za neki realni broj  $u$ .

U programu *Mathematica* te točke i težište  $T$  zadajemo ovako:

$tA := \{0, 0\}; tB := \{c, 0\}; tC := \{u, v\}; tT := \text{teziste}[tA, tB, tC]$


Budući da su težišnice iz vrhova  $A$  i  $B$  okomite, trokut  $ABT$  je pravokutan pa vrijedi  $c^2 = |AB|^2 = |AT|^2 + |BT|^2$ . S druge strane imamo  $|BC| = 6$  i  $|AC| = 8$ . Zatražimo li od programa *Mathematica* da riješi taj sustav od tri jednačbe u nepoznicama  $c$ ,  $u$ , i  $v$  unosom

```
Solve[{udaljenost2t[tB,tC]==6, udaljenost2t[tA,tC]==8,
c^2==udaljenost2t[tA,tT]^2+udaljenost2t[tB,tT]^2},{c,u,v}]
```

on će ponuditi četiri rješenja od kojih samo  $c = 2\sqrt{5}$  cm odgovara.  $\square$

Naš slijedeći primjer je Zadatak 1112. iz [4].

**Zadatak 13.** U trapez je upisan krug. Dokaži da je omjer površina trapeza i kruga jednak omjeru njihovih opsega.



SLIKA 18. Omjeri površina i opsega kruga i njemu opisanog trapeza su isti.

*Rješenje.* Odaberimo pravokutni koordinatni sustav tako da kružnica  $k$  radijusa  $R$  koja je upisana promatranom trapezu  $ACEG$  ima središte u ishodištu, a njegove paralelne stranice  $AC$  i  $EG$  dodiruju  $k$  u točkama  $B(0, -R)$  i  $F(0, R)$ . Neka vrhovi  $A$  i  $C$  imaju koordinate  $(-u, -R)$  i  $(v, -R)$  za pozitivne realne brojeve  $u$  i  $v$ . Neka krakovi  $CE$  i  $AG$  dodiruju  $k$  u točkama  $D$  i  $H$ . Naš prvi cilj je odrediti koordinate tih dirališta, a zatim koordinate vrhova  $E$  i  $G$ .

Zadajmo u programu *Mathematica* prvo točke  $O$ ,  $B$ ,  $F$ ,  $A$ ,  $C$  i pravce  $AC$ ,  $EG$ .

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R};
tA:={-u, -R}; tC:={v, -R};
pAC:={0, 1, R}; pEG:={0, 1, -R}
```

Pretpostavimo da diralište  $H$  ima koordinate  $(p, q)$ . One moraju zadovoljavati dva uvjeta. Prvi je  $p^2 + q^2 = R^2$  tj. da  $H$  leži na kružnici  $k$ .

Drugi uvjet je da je udaljenost od  $A$  do  $H$  jednaka  $u$  jer su pravci  $AB$  i  $AH$  dvije tangente iz točke  $A$  na kružnicu  $k$ .

```
H:=Solve[{p^2+q^2==R^2, udaljenost2t[{p, q}, tA]==u}, {p, q}]
tH:={p, q} /. Extract[H, 2]
```

Slično se dobivaju i koordinate dirališta  $D$ .

```
K:=Solve[{p^2+q^2==R^2, udaljenost2t[{p, q}, tC]==v}, {p, q}]
tD:={p, q} /. Extract[K, 2]
```

Vrhovi  $E$  i  $G$  su presjeci pravca  $EG$  sa pravcima  $CD$  i  $AH$ .

```
pAH:=pravac2t[tA, tH]; pCD:=pravac2t[tC, tD];
tE:=presjek2p[pEG, pCD]; tG:=presjek2p[pEG, pAH]
```

Prve koordinate točaka  $E$  i  $G$  su  $\frac{R^2}{v}$  i  $-\frac{R^2}{u}$ . Zato je opseg  $O_{ACEG}$  trapeza  $ACEG$  jednak  $2(u + v + \frac{R^2}{u} + \frac{R^2}{v})$ . Njegova površina  $P_{ACEG}$  je

```
povrsina3t[tA, tC, tE]+povrsina3t[tA, tE, tG]
```

jednaka  $\frac{R(u+v)(u+v+R^2)}{uv}$ . Sada se lagano provjeri da je


$$\frac{O_{ACEG}}{2R\pi} = \frac{P_{ACEG}}{R^2\pi}.$$

□

*Napomena 4.* U knjizi [4] nema rješenja za zadatak 1112.

Naš idući primjer je Zadatak 1139. iz [4].

**Zadatak 14.** Dokaži, ako je pravac na kojemu leži simetrala kuta trokuta ujedno i simetrala kuta pravaca na kojima leže visina i težišnica povučene iz vrha tog kuta, onda je trokut pravokutan.



SLIKA 19. Trokut je ili jednakokračan ili pravokutan ako simetrala kuta raspolavlja kut između visine i težišnice.

*Rješenje.* Odaberimo pravokutni koordinatni sustav tako da su vrhovi točke  $A(0, 0)$ ,  $B((f+g)r, 0)$ ,  $C\left(\frac{rg(f^2-1)}{fg-1}, \frac{2fgr}{fg-1}\right)$ , a središte upisane kružnice je točka  $I(fr, r)$ , gdje su  $f$  i  $g$  kotangensi od  $\frac{A}{2}$  i  $\frac{B}{2}$ , a  $r$  je radijus upisane kružnice (vidi rješenje Zadatka 6).

Zadajmo u programu *Mathematica* prvo točke  $A$ ,  $B$ , polovište  $C_g$  dužine  $AB$ , točke  $C$ ,  $I$  i nožište  $C_h$  visine iz vrha  $C$  na pravac  $AB$ .

```
tA:={0, 0}; tB:={r*(f+g), 0}; tCg:=poloviste[tA,tB];
tC:={r*g*(f^2-1)/(f*g-1), 2*f*g*r/(f*g-1)};
tI:={f*r, r}; tCh:=projekcija[tC,pravac2t[tA,tB]]
```

Da bi simetrala kuta  $C$  (tj. pravac  $CI$ ) bio simetrala kuta određenog visinom (tj. pravcem  $CC_h$ ) i težišnicom (tj. pravcem  $CC_g$ ) nužno je i dovoljno da dužine  $II_h$  i  $II_g$  imaju istu duljinu, gdje su  $I_h$  i  $I_g$  ortogonalne projekcije točke  $I$  na pravce  $CC_h$  i  $CC_g$ .

```
tIh:=projekcija[tC,pravac2t[tC,tCh]];
tIg:=projekcija[tC,pravac2t[tC,tCg]];
IZ:=FS[udaljenost2t[tI,tIg]^2-udaljenost2t[tI,tIh]^2]
```

Program *Mathematica* izvještava da je izraz  $IZ$  jednak

$$\frac{r^2(f-g)^2(fg+g+f-1)(fg-g-f-1)(fg+1)^2(fg-1)^{-2}}{(12f^2g^2+g^2f^4-2f^3g^3+g^4f^2+2f^3g+2fg^3+f^2-2fg+g^2)}.$$

Dakle, on će biti nula onda i samo onda ako je  $f = g$  (tj.  $|BC| = |CA|$  pa je trokut  $ABC$  jednakokračan) ili je

$$(fg+g+f-1)(fg-g-f-1) = 0$$

što je upravo uvjet da pravci  $BC$  i  $CA$  budu okomiti (tj. da je kut  $C$  pravi pa je trokut  $ABC$  pravokutan).

```
okomitiQ[pravac2t[tB, tC], pravac2t[tC, tA]]
```

□

*Napomena 5.* U knjizi [4] ne spominje se mogućnost da trokut  $ABC$  bude jednakokračan.

Naš posljednji primjer je Zadatak 1152. iz [4].

**Zadatak 15.** Zadane su točke  $A$  i  $B$  i točka  $T$  koja ne leži na pravcu  $AB$ . Točkom  $T$  povuci pravac  $m$  tako da je omjer udaljenosti točaka  $A$  i  $B$  od  $m$  jednak  $2:3$ .


*Rješenje.* Odaberimo pravokutni koordinatni sustav tako da je  $A(0, 0)$ ,  $B(c, 0)$ , i  $T(p, q)$ . Neka pravac  $m$  ima jednadžbu  $ux + vy + w = 0$ . Da bi on prolazio točkom  $T$  mora biti  $w = -up - vq$ .

Zadajmo u programu *Mathematica* točke  $A$ ,  $B$ ,  $T$  i pravac  $m$ .

```
tA:={0, 0}; tB:={c, 0}; tT:={p, q}; pm:={u, v, -u*p-v*q}
```

Neka su  $A_m$  i  $B_m$  ortogonalne projekcije točaka  $A$  i  $B$  na pravac  $m$ .

```
tAm:=projekcija[tA, pm]; tBm:=projekcija[tB, pm]
```



SLIKA 20. Postoje dva pravca  $a$  i  $b$  točkom  $T$  za koje su omjeri udaljenosti točaka  $A$  i  $B$  do njih jednaki  $\frac{2}{3}$ .

Prema uvjetu zadatka mora biti  $\frac{|AA_m|}{|BB_m|} = \frac{2}{3}$ . Primjetimo da izraz

$$IZ := FS[udaljenost2t[tA, tAm]^2 / udaljenost2t[tB, tBm]^2 - 4/9]$$

ima za brojnik produkt  $(5up + 5vq - 2uc)(up + vq + 2uc)$ . Dakle, imamo dvije mogućnosti  $v = \frac{-u(2c+p)}{q}$  i  $v = \frac{u(2c-5p)}{5q}$ . Tako dobivamo dva rješenja pravce  $qx - (2c + p)y + 2qc = 0$  i  $5qx + (2c - 5p)y - 2qc = 0$ . Iako znamo rješenja preostaje pitanje kako te pravce lakše odrediti. Ali, vidimo da oni sijeku pravac  $AB$  u točkama s koordinatama  $C(-2c, 0)$  i  $D(\frac{2}{5}c, 0)$  koje je lagano konstruirati.  $\square$

*Napomena 6.* U knjizi [4] nema rješenja za zadatak 1152.

## REFERENCE

- [1] M. Bator, Z. Čerin i M. Čulav, *Analitička geometrija ravnine računalom*, Matematičko-fizički list, (prijavljeno).
- [2] Zvonko Čerin, *Tucet zadataka riješenih programom Maple V*, Matematičko-fizički list, (prijavljeno).
- [3] Zvonko Čerin, *Geometrija pravilnog sedmerokuta*, Poučak (i/ili Playmath), (prijavljeno).
- [4] Boris Pavković i Darko Veljan, *Matematika 1 – Zbirka zadataka s uputama i rješenjima za prvi razred srednjih škola (14. dopunjenjo izdanje)*, Školska knjiga, Zagreb, 1999.

ODRANSKA 8, 10000 ZAGREB, HRVATSKA, EUROPA  
E-mail address: batormaja@yahoo.com

KOPERNIKOVA 7, 10020 ZAGREB, HRVATSKA, EUROPA  
E-mail address: cerin@math.hr

LHOTKINA 3, 10000 ZAGREB, HRVATSKA, EUROPA  
E-mail address: mickey@student.math.hr